Laboratory 1: Solving differential equations with MAPLE

Functions and graphic representation

A single variable function can be defined as follows:

```
> f:=x->sin(x)/x;
```

$$f := x \to \frac{\sin(x)}{x}$$

```
> f(3*Pi/2),f(1.5);
```

$$-\frac{2}{3\pi}, 0.6649966577$$

```
> f(a+b);
```

$$\frac{\sin(a+b)}{a+b}$$

For graphical representation we need to load **plots** package using **with** command

```
> with(plots):
Warning, the name changecoords has been redefined
```

```
> plot(f(x),x=-1..1);
```



We can represent more than one function in the same window:

```
> plot([f(x),f(2*x),sin(x)],x=0..2*Pi,color=[red,blue,green]);
```



In the case of discontinuous points we need to use the option **discont = true**:

```
> plot(tan(x), x = -2*Pi..2*Pi, y = -4..4, discont = true);
```



If a curve is given in a parametric form (for example: $x(t) = \sin(t)$, $y(t) = \cos(t)$, $t = 0 .. \pi$;) we use the instruction:

```
> plot([sin(t),cos(t),t=0..3/2*Pi]);
```



In the case of a curve given by the implicit equation we use the instruction **implicitplot**:

```
> implicitplot(x^2+y^2=1,x=-1..1,y=-1..1);
```



In the case we need to visualize the parameter dependence of a function we can use the command **animate**

(right click on the image, select *Animation* and *Play*)

```
> animate(sin(x*t),x=-4*Pi..4*Pi,t=0..2,color=red);
```



If we need more precision we can increase the number of point and frames:

```
> animate(sin(x*t),x=-4*Pi..4*Pi,t=0..2,color=red,numpoints=100,frames=100);
```



A function with more than one variable can be defined as follows:

```
> g:=(x,y)->sin(x)*exp(-y);
```

$$g := (x, y) \to \sin(x)\, \mathbf{e}^{(-y)}$$

For 3d graphical representation we use the command **plot3d**:

```
> plot3d(g(x,y),x=0..Pi,y=0..3,axes=boxed);
```



The animation of 3D graphs can be made using the instruction **animate3d**:

> `animate3d(g(t*x,y),x=0..Pi,y=0..3,t=0..2);`



The derivation of the functions

The derivation of the functions can be made in two ways: using **diff** command or using the derivation operator **D**:

> `f:=x->exp(x^2)+3;`

$$f := x \rightarrow \mathbf{e}^{(x^2)} + 3$$

The **diff** command execute the derivation of the given expresion with respect to the specified variable. The derivation operator **D** returns the derivate as a function.

> `diff(f(x),x);`

$$2\,x\,\mathbf{e}^{(x^2)}$$

the second order derivate is given by

> `diff(f(x),x,x);`

$$2\,\mathbf{e}^{(x^2)} + 4\,x^2\,\mathbf{e}^{(x^2)}$$

also we can use the option **x$n** to get n-order derivative

> `diff(f(x),x$2);`

$$2\,\mathbf{e}^{(x^2)} + 4\,x^2\,\mathbf{e}^{(x^2)}$$

> `diff(f(x),x$3);`

$$12\,x\,\mathbf{e}^{(x^2)} + 8\,x^3\,\mathbf{e}^{(x^2)}$$

Using the derivation operator:

> `D(f)(x);`

$$2\,x\,\mathbf{e}^{(x^2)}$$

> `D(f)(1);`

$$2\,\mathbf{e}$$

> `(D@D)(f)(x);`

$$2\,\mathbf{e}^{(x^2)} + 4\,x^2\,\mathbf{e}^{(x^2)}$$

```
> (D@D)(f)(1);
```

$$6\,e$$

```
> (D@@2)(f)(x);
```

$$2\,e^{(x^2)} + 4\,x^2\,e^{(x^2)}$$

```
> (D@D@D)(f)(x);
```

$$12\,x\,e^{(x^2)} + 8\,x^3\,e^{(x^2)}$$

```
> (D@@3)(f)(x);
```

$$12\,x\,e^{(x^2)} + 8\,x^3\,e^{(x^2)}$$

Initialization of the solving ODE package

```
> restart:
```
clears the memory of all previously saved values and variables

```
> with(DEtools):
```
load the differential equations package

```
> with(plots):
```
load the graphical package

```
Warning, the name changecoords has been redefined
```

Define and solve a first order differential equation

Let consider the differential equation $\frac{d}{dx}\,y(x) = k\,y(x)$ where k is a real coeficient. The differential equation can be introduse in MAPLE as follows:

```
> diff_eq1:=diff(y(x),x) = k*y(x);
```

$$diff\_eq1 := \frac{d}{dx}\,y(x) = k\,y(x)$$

To obtain the general solution of the equation use dsolve command

```
> dsolve(diff_eq1,y(x));
```

$$y(x) = \_C1\,e^{(k\,x)}$$

The general solution is seen as an expresion. Notice that the undetermined constant is called *_C1*
How can we manipulate this expresion?
We can use the function definition command:

```
> sol:=(x,k,c)->c*exp(k*x);
```

$$sol := (x, k, c) \rightarrow c\,e^{(k\,x)}$$

If the expresion of the solution is too complicated we can use the command **rhs** (*right hand side*) and **unapply** in order to obtain the solution as a function

```
> right_hand_expr:=rhs(dsolve(diff_eq1,y(x)));
```

$$right\_hand\_expr := \_C1\,e^{(k\,x)}$$

Using the **unapply** command we transform the expression sol1 into a function specifying the variables:

```
> sol1:=unapply(right_hand_expr,x,k,_C1);
```

$$sol1 := (x, k, \_C1) \rightarrow \_C1\,e^{(k\,x)}$$

and we get the same result.

The graphics of ODE solutions

Let suppose that k:=2. Then the corresponding general solution is:

```
> y:=(x,c)->sol(x,2,c);
```

$$y := (x, c) \rightarrow \mathrm{sol}(x, 2, c)$$

To draw the solutions curves you just assign some values for the constant c. For example take $c := 1$ $c := 2$ and $c := -1$

```
> plot([y(x,1),y(x,2),y(x,-1)],x=-2..2);
```



If you want to obtain the solutions with some specified colors use the command:

```
> plot([y(x,1),y(x,2),y(x,-1)],x=-2..2,color=[black,red,blue]);
```



Also you specify the window of the graphic:

```
> plot([y(x,1),y(x,2),y(x,-1)],x=-2..2,y=-10..10,color=[black,red,blue]);
```



Using this way of manipulation for the solution you can see also how the solutiond depends on the k parameter. Let us consider $c := 1$ and assign some values for th parameter k.

```
> y1:=(x,k)->sol(x,k,1);
```

$$y1 := (x, k) \to \mathrm{sol}(x, k, 1)$$

```
> plot([y1(x,0.2),y(x,0.5),y(x,-1)],x=-2..2,y=-10..10,color=[black,red,blue]);
```



Testing solutions

For a given differential equation we can test if some given function satisfies or does not satisfy that differential equation using **odetest** command. If **odetest** returns **0** then the given function is a solution, if **odetest** returns something **different from 0** then the given function is not a solution. Let us consider the the differential equation $\frac{d}{dx}\mathrm{y}(x) = 2\,\mathrm{y}(x)$ and check if $\mathrm{y1}(x) = 4\,\mathrm{e}^{(2\,x)}$, $\mathrm{y2}(x) = \mathrm{e}^{(3\,x)}$ and $\mathrm{y3}(x) = \cos(2\,x)$ are solution for this equation.

```
> restart:with(DEtools):
> diff_eq:=diff(y(x),x) = 2*y(x);
```

$$diff\_eq := \frac{d}{dx}\mathrm{y}(x) = 2\,\mathrm{y}(x)$$

```
> y1:=x->4*exp(2*x);y2:=x->exp(3*x);y3:=x->cos(2*x);
```

$$y1 := x \to 4\,\mathrm{e}^{(2\,x)}$$

$$y2 := x \to \mathrm{e}^{(3\,x)}$$

$$y3 := x \to \cos(2\,x)$$

```
> odetest(y(x)=y1(x),diff_eq,y(x));
```

$$0$$

for $y1(x) = 4\,e^{(2\,x)}$ **odetest** returns **0**, so $y1(x)$ is a solution of this equation

```
> odetest(y(x)=y2(x),diff_eq,y(x));
```

$$e^{(3\,x)}$$

for $y2(x) = e^{(3\,x)}$ **odetest** returns $e^{(3\,x)}$ which is not **0**, so $y2(x)$ is not a solution of the equation, in fact **odetest** evaluates $\left(\frac{d}{dx}\,y(x)\right) - 2\,y(x)$ for $y(x) = y2(x)$, therefore if we get **0** this means that the function is a solution, if the result is something different from **0** then the function is not a solution for the differential equation.

```
> odetest(y(x)=y3(x),diff_eq,y(x));
```

$$-2\,\sin(2\,x) - 2\,\cos(2\,x)$$

also $y3(x)$ is not a solution since the result of **odetest** is different from **0.**

Solving an IVP

Suppose that we want to solve the IVP $\frac{d}{dx}\,y(x) = k\,y(x)$ with the initial condition $y(0) := 1$

```
> restart:with(DEtools):
> diff_eq:=diff(y(x),x) = k*y(x);
```

$$diff\_eq := \frac{d}{dx}\,y(x) = k\,y(x)$$

```
> in_cond:=y(0)=1;
```

$$in\_cond := y(0) = 1$$

```
> dsolve({diff_eq,in_cond},y(x));
```

$$y(x) = e^{(k\,x)}$$

Let consider the case $k := 2$

```
> k:=2;
```

$$k := 2$$

```
> sol:=dsolve({diff_eq,in_cond},y(x));
```

$$sol := y(x) = e^{(2\,x)}$$

```
> yy:=x->(rhs(sol));
```

$$yy := x \to \mathrm{rhs}(sol)$$

```
> plot(yy(x),x=-1..1,y=0..4);
```



You can obtain the graph the IVP directly using the command DEplot:

```
> DEplot(diff_eq,y(x),x=-1..1,[[in_cond]]);
```



In this graph is also represented the direction field of the equation. If you want the graphs of the solutions for different initial condition ($y(0)=1$, $y(0)=1.5$, $y(0)=2$) you can use the same command and specify the list of initial conditions:

```
> DEplot(diff_eq,y(x),x=-1..1,[[y(0)=1],[y(0)=1.5],[y(0)=2]]);
```



Solving a second order ODE

```
> restart:
> with(DEtools):
> with(plots):
Warning, the name changecoords has been redefined
```

Consider the linear differential equation with the constant coeficients

$$\left(\frac{d^2}{dx^2}\mathrm{y}(x)\right)+3\left(\frac{d}{dx}\mathrm{y}(x)\right)+2\,\mathrm{y}(x)=1+x^2$$

```
> deq1:=diff(y(x),x$2)+3*diff(y(x),x)+2*y(x)=1+x^2;
```

$$deq1 := \left(\frac{d^2}{dx^2}\mathrm{y}(x)\right)+3\left(\frac{d}{dx}\mathrm{y}(x)\right)+2\,\mathrm{y}(x)=1+x^2$$

To obtain the general solution we use the dsolve command

```
> dsolve(deq1,y(x));
```

$$\mathrm{y}(x)=\frac{9}{4}+\frac{x^2}{2}-\frac{3\,x}{2}-\mathbf{e}^{(-2\,x)}\,\_C1+\mathbf{e}^{(-x)}\,\_C2$$

If we want to study the solution we can use the same technique as in the previous section in order to draw the solution graph.

```
> sol:=dsolve(deq1,y(x));
```

$$sol := \mathrm{y}(x)=\frac{9}{4}+\frac{x^2}{2}-\frac{3\,x}{2}-\mathbf{e}^{(-2\,x)}\,\_C1+\mathbf{e}^{(-x)}\,\_C2$$

```
> right_hand:=rhs(sol);
```

$$right\_hand := \frac{9}{4}+\frac{x^2}{2}-\frac{3\,x}{2}-\mathbf{e}^{(-2\,x)}\,\_C1+\mathbf{e}^{(-x)}\,\_C2$$

```
> y_sol:=unapply(right_hand,x,_C1,_C2);
```

$$y\_sol := (x, \_C1, \_C2) \rightarrow \frac{9}{4}+\frac{1}{2}x^2-\frac{3}{2}x-\mathbf{e}^{(-2\,x)}\,\_C1+\mathbf{e}^{(-x)}\,\_C2$$

Now we are able to one ore more than one solution graphs using the **plot** command.

```
> plot([y_sol(x,0,0),y_sol(x,0,1),y_sol(x,1,0)],x=-2..2,y=-10..10);
```



In the case of initial value problem we have two initial conditions, for example lets take $\mathrm{y}(0)=1$ and y'(0)=0.

```
> in_cond:=y(0)=1,D(y)(0)=0;
```

$$in\_cond := \mathrm{y}(0)=1,\ \mathrm{D}(y)(0)=0$$

To obtain the corresponding solution we use **dsolve** command in the following form:

```
> dsolve({deq1,in_cond},y(x));
```

$$y(x) = \frac{9}{4} + \frac{x^2}{2} - \frac{3\,x}{2} - \frac{1}{4}\,e^{(-2\,x)} - e^{(-x)}$$

Now we can use the previous technique (**rhs** and **unapply** comands) to construct the solution as a function and after that to represent its graph or we can obtain this graph directly using **DEplot** command.

```
> DEplot(deq1,y(x),x=-2..2,y=-10..10,[[in_cond]]);
```



If we need to draw more than one solution corresponding to different initial value problem we can use the same DEplot command specifying the list of initial conditions:

```
> DEplot(deq1,y(x),x=-
2..2,[[y(0)=1,D(y)(0)=0],[y(0)=1,D(y)(0)=1],[y(0)=0,D(y)(0)=0]],y=-
10..10,linecolor=[red,blue,green]);
```



The general second order linear DE, $\left(\frac{d^2}{dx^2}\,y(x)\right) + p(x)\left(\frac{d}{dx}\,y(x)\right) + q(x)\,y(x) = f(x)$:

Note: Maple is unable to solve most second-order DE's explicitly. For information on numerically solving DE's, see Numerical Solutions with dsolve.

Consider the differential equation $\left(\frac{d^2}{dx^2}\,y(x)\right) + x\left(\frac{d}{dx}\,y(x)\right) + y(x) = \sin(x)$. Try to use the **dsolve** command.

```
> deq2:=diff(y(x),x$2)+x*diff(y(x),x)+y(x)=sin(x);
```

$$deq2 := \left(\frac{d^2}{dx^2}\,y(x)\right) + x\left(\frac{d}{dx}\,y(x)\right) + y(x) = \sin(x)$$

```
> dsolve(deq2,y(x));
```

$$y(x) = e^{\left(-\frac{x^2}{2}\right)}\,\_C1\,\operatorname{erf}\left(\frac{1}{2} I \sqrt{2}\, x\right) + e^{\left(-\frac{x^2}{2}\right)}\,\_C2$$
$$+ \frac{1}{4} I \sqrt{2}\,\sqrt{\pi}\, e^{(1/2)} \left(\operatorname{erf}\left(\frac{1}{2} I \sqrt{2}\, x - \frac{\sqrt{2}}{2}\right) + \operatorname{erf}\left(\frac{1}{2} I \sqrt{2}\, x + \frac{\sqrt{2}}{2}\right)\right) e^{\left(-\frac{x^2}{2}\right)}$$

```
> in_cond2:=y(0)=1,D(y)(0)=1;
```

$$in\_cond2 := y(0) = 1,\ D(y)(0) = 1$$

```
> dsolve({deq2,in_cond2},y(x));
```

$$y(x) = -e^{\left(-\frac{x^2}{2}\right)}\sqrt{\pi}\,\sqrt{2}\,\operatorname{erf}\left(\frac{1}{2} I \sqrt{2}\, x\right) I + e^{\left(-\frac{x^2}{2}\right)}$$
$$+ \frac{1}{4} I \sqrt{2}\,\sqrt{\pi}\, e^{(1/2)} \left(\operatorname{erf}\left(\frac{1}{2} I \sqrt{2}\, x - \frac{\sqrt{2}}{2}\right) + \operatorname{erf}\left(\frac{1}{2} I \sqrt{2}\, x + \frac{\sqrt{2}}{2}\right)\right) e^{\left(-\frac{x^2}{2}\right)}$$

Maple expresses the solution in terms of the error function **erf**.
We can obtain the numerical solution using in the dsolve command the option ' *type* = *numeric* ' and the odeplot comand to draw the corresponding graph.

```
> n_sol:=dsolve({deq2,in_cond2},y(x),type=numeric):
> odeplot(n_sol);
```